
  
  
  

 
 
By Melinda Gallo, Code by Dave Batton.  
Technical Note 05-24 
 
（原題: Client Self-Upgradable） 
 
概要概要  
 
4D Internet Commands プラグインの SMTP_Auth コマンドを使用すれば、認証を要求す
る SMTP サーバを通して 4D から電子メールを送信することができます。これに対して
SMTP_QuickSendコマンドは手軽な反面、認証を扱うことができません。今回、紹介する
Email_QuickSend プロジェクトメソッドは、SMTP_QuickSend コマンドに認証のための
パラメータを追加したものです。 
 
サンプルデータベース 
 
Email_Demoを起動して、「ファイル」メニューから「Demo」を選択します。 
 

 
 
Hostname には、SMTPサーバの名称を入力します。 
Sender には、送信者のメールアドレスを入力します。 
パスワードの入力は Sendボタンをクリックした後に求められます。 
Recipient には宛先のアドレス、Subject には件名、Message には本文を入力します。 
 
すべて必須項目です。 

Sendボタンをクリックすると、次の画面が表示されます。 
 

 
 
パスワードは SMTPサーバによって検証され、正しければ、メールが伝達されます。 
 
ストラクチャ 
 
スタラクチャには“SendEmail”というフォームがひとつだけ存在します。 
 

 
 
フォームメソッド 
 
Case of  
 ¥ (Form event=On Load ) 
  C_STRING(80;vnomHost;vSender;vRecipiend;vSubject) 
  C_TEXT(vmessage) 
  vnomHost:="" 
  vSender:="" 
  vRecipiend:="" 
  vSubject:="" 
  vmessage:="Enter a message here ." 
End case 
 
Send ボタンのオブジェクトメソッド 



 
Case of  
 ¥ (vnomHost="") 
  ALERT("Please enter a hostname.") 
 ¥ (vSender="") 
  ALERT("Please enter an email address.") 
 ¥ (vRecipiend="") 
  ALERT("Please enter a recipiend address.") 
 ¥ (vSubject="") 
  ALERT("Plase enter a subject.") 
 ¥ (vmessage="") 
  ALERT("Please enter a message.") 
 Else  
  mSendEmail(vnomHost;vSender;vRecipiend;vSubject;vmessage) 
End case 
 
すべての項目を確認した後、mSendMail をコールしています。このメソッドのシンタック
スは、パスワードの引数があることを除けば、SMTP_QuickSendとまったく同じです。 
 
C_STRING(80;$1;$2;$3;$4;$Hostname;$Sender;$Recipiend;$subject) 
C_TEXT($5;$message) 
 
$Hostname:=$1 
$Sender:=$2 
$Recipiend:=$3 
$subject:=$4 
$message:=$5 
$password:=Request("Enter your password: ") 
 
If (OK=1) 
 $error:=Email_QuickSend($Hostname;$Sender;$Recipiend;$subject;$message;$password) 
End if 
 
Email_QuickSend 
 
Email_QuickSend (hostname; Sender; recipient; subject; message; password) 
 

引数 タイプ 説明 
Hostname 文字 SMTPサーバの名前 
Sender 文字 送信者のアドレス 
recipient 文字列 宛先のアドレス 
subject 文字 件名 
message テキスト 本文 
password 文字 送信者アカウントのパスワード 
 
メソッドは、SMTP_QuickSendの代わりにローレベルコマンド（SMTP_New, SMTP Host, 
SMTP_Auth, etc.）を使用しています。サーバが認証を要求しないなら SMTP_QuickSend
コマンドを使用したほうが便利ですが、正しく運用しないとオープンリレーのような状態
になり、セキュリティ違反になる可能性があるので注意が必要です。 
 
C_LONGINT($0;Error;$smtpID) 



C_TEXT($1;$2;$3;$4;$5;$6;$smtpServer;$Sender;$Recipiend;$subject;$message;$password) 
 
$smtpServer:=$1 
$Sender:=$2 
$Recipiend:=$3 
$subject:=$4 
$message:=$5 
$password:=$6 
 
Case of  
 ¥ (Not(Email_HandleSMTPError ("SMTP_New";SMTP_New ($smtpID)))) 
 ¥ (Not(Email_HandleSMTPError ("SMTP_Host";SMTP_Host ($smtpID;$smtpServer))))  ` 
 ¥ (Not(Email_HandleSMTPError ("SMTP_Auth";SMTP_Auth ($smtpID;$Sender;$password)))) 
 ¥ (Not(Email_HandleSMTPError ("SMTP_From";SMTP_From ($smtpID;$Sender)))) 
 ¥ (Not(Email_HandleSMTPError ("SMTP_To";SMTP_To ($smtpID;$Recipiend)))) 
 ¥ (Not(Email_HandleSMTPError ("SMTP_Subject";SMTP_Subject ($smtpID;$subject)))) 
 ¥ (Not(Email_HandleSMTPError ("SMTP_Body";SMTP_Body ($smtpID;$message;2)))) 
 Else  
  Email_HandleSMTPError ("SMTP_Send";SMTP_Send ($smtpID)) 
  If ($smtpID#0) 
   Email_HandleSMTPError ("SMTP_Clear";SMTP_Clear ($smtpID)) 
  End if  
End case  
 
$0:=Error  ` Always returns 0. 
 
Email_HandleSMTPError 
 
Email_HandleSMTPError (command; error) -> Boolean 
 

引数 タイプ 説明 
Command テキスト 実行するコマンドの名前 
Error 倍長整数 エラー番号 
 
このメソッドは、4D Internet Commandの各コマンドにつきメソッドEmail_QuickSend
の中でコールされます。エラーが発生した場合、コマンド名とエラー番号が警告ダイアロ
グに表示されます。エラーがなければTrue を返し、そうでなければ False を返します。 
 
C_BOOLEAN($0) 
C_TEXT($1;$command) 
C_LONGINT($2;$error) 
 
$command:=$1  `Command name 
$error:=$2  `Error number (zero if no error) 
 
If ($error=0)  `no d'error 
 $0:=True 
Else   `an error took place 
 $0:=False 
 Error:=$error  `returning error number in system Error 
 ALERT("The error "+String($error)+" was returned by the command "+$command) 
End if 


